
Recreating historical patterns with MetaPost

Mari Voipio

I’ve always been interested in history, and fascinated
by repeated geometric patterns. This combination
of interests has led me to try to recreate patterns
from various historical periods for use in crafts, e.g.
card making and quilting. Other patterns presented
here were drawn just because they intrigued me and
looked like a good way of practicing with MetaPost.
I tend to prefer older European history, but the same
motifs recur in later periods, so many look familiar —
even modern — like the Greco-Roman pelta.

1 The Greek key

The Greek key and its variations have a multitude of
names: meander, scroll, running dog, . . . The Greek
key appears all over the Classical world, being very
easy to repeat e.g. in mosaic. The basic square or
rectangular form is also easy to recreate with Meta-
Post and allows for play with borders and positive/
negative images.

% define one unit

picture gkey;

gkey := image (

linecap := squared ; % crisp triangle corners

linejoin := mitered ; % neat line ends

draw (0,0) -- (0,5) -- (5,5) -- (5,1) -- (2,1)

-- (2,3) -- (3,3) -- (3,2) -- (4,2)

-- (4,4) -- (1,4) -- (1,0) -- (6,0);

);

% repeat to create a border

for i = 0 step 6 until 24 :

draw gkey shifted (i,0) ;

endfor ;

% set final size of the meander

currentpicture := currentpicture xsized 6cm ;

% add .5 cm whitespace all over

setbounds currentpicture

to boundingbox currentpicture enlarged 1/2cm ;

This “reversed” image draws the meander in
white on a black background, the latter made with:

addbackground withcolor black ;

2 Roman mosaic floors

The Roman villas were richly decorated with paint-
ings and mosaic floors — not only in Italy, but any-
where the Romans went, including the British Isles.
The mosaics are laid with small cubes, tesserae, that
are cut to points when needed. However, I’m more
interested in the patterns that form when the mosaic
has been laid. My first attempt at recreating mosaic
patterns was a monochromatic floor border found in
an older layer (1st century) at Fishbourne Roman
villa in the south of England.

This set of patterns provided me with an ad-
mirable opportunity to explore MetaPost’s capabili-
ties, because each subpattern is divided in different
numbers of units: 9, 8, 10. I dreaded doing the math
until I realized that MetaPost can do fractions just
as easily as decimals, so drawing everything in the
same size was no big deal.

Later I learned that it might be easier to use
whole numbers and then scale the resulting square
elements into the same width, but there’s a certain
beauty in creating the pattern just as the Romano-
Britain masons would have, by first dividing the area
into squares, then dividing the squares into as many
equal parts as needed for the pattern.

Creating the wheel (third) pattern:

linejoin := mitered ;

path whtrcomp ; % triangle for the wheel pattern

whtrcomp = ( % on a 5-step grid

(0,0) -- (2/5,1/5) -- (1/5,2/5)

-- (0,0) -- cycle ) ;

path whsqcomp; % square for the wheel pattern

whsqcomp = (

(2/5,1/5) -- (4/5,0) -- (5/5,2/5)

-- (3/5,3/5) -- (2/5,1/5) -- cycle ) ;

% tiling the triangles and squares

% into a mosaic block of 2+2

picture whmosaic;

whmosaic := image (

fill whtrcomp;

fill whsqcomp ;

fill whsqcomp reflectedabout ((0,0),(1,1)) ;

fill whtrcomp shifted (3/5,3/5) ;

);

% tiling the quarter blocks into a wheel

picture wheel ;

wheel := image (

draw whmosaic;

draw whmosaic rotatedaround ((1,1),90);

draw whmosaic rotatedaround ((1,1),180);

draw whmosaic rotatedaround ((1,1),270);

) ;

% drawing the wheel pattern

draw wheel ;

% one way to add a frame around the wheel

setbounds currentpicture

to boundingbox currentpicture enlarged 1/10 ;

draw boundingbox currentpicture

withpen pencircle scaled 1/5 ;

% setting the size of the final graphic

currentpicture := currentpicture xsized 2.6cm ;

The pelta— shield shape — is a frequently re-
peated pattern in early art, even on floor mosaics,
despite its curvy nature being a bit of a challenge
when laying the mosaic. I, too, had to fight with
the pattern for a while before I realized that my
curves behave better if I add a nil-length straight
line between the curves.

The guest section of Hadrian’s villa in Tivoli,
Italy, has a floor patterned with mirrored and ro-
tated pelta and squares (Field, 1988, p. 62). The con-
stituent parts are simple to code, but the combined
floor pattern turned out to be harder to achieve than
I first thought. Somehow the rotated and shifted and
repeated curves seemed to mislead my eyes, and only
after several tries did my graphic match the floor.

% defining a single pelta

picture pelta ;

pelta := image (

linejoin := mitered;

linecap := squared ;

filldraw (0,0) .. (2,2) .. (4,0) -- (4,0)

.. (3,1) .. (2,0) -- (2,0) .. (1,1)

.. (0,0) -- cycle

withpen pencircle scaled 1/50; ) ;

% defining a block of two

% square+pelta combinations

picture flooring;

flooring := image (

draw pelta shifted (-1,1);

draw pelta rotated 90 shifted (-1,-1);

draw pelta rotated 180 shifted (1,-1);

draw pelta rotated 270 shifted (1,1);

fill fullsquare scaled 2;

draw pelta shifted (3,1);

draw pelta rotated 90 shifted (5,-3);

draw pelta rotated 180 shifted (9,-1);

draw pelta rotated 270 shifted (7,3);

fill fullsquare scaled 2 shifted (6,0);

);

draw flooring reflectedabout

((ulcorner flooring),(urcorner flooring));

% rows 1 and 3

for i = 0 step 12 until 18 :

for j = 0 step 12 until 18 :

draw flooring shifted (i,j) ;

endfor ;

endfor ;

% rows 2 and 4

for i = 0 step 12 until 18 :

for j = 0 step 12 until 18 :

draw flooring reflectedabout

((ulcorner flooring),(urcorner flooring))

shifted (i,j);

endfor ;

endfor ;

% adding the borderline around the flooring

% (as in the original floor)

setbounds currentpicture

to boundingbox currentpicture enlarged 1/10 ;

draw boundingbox currentpicture

withpen pencircle scaled 1/5 ;

% adding whitespace around the picture

setbounds currentpicture

to boundingbox currentpicture enlarged 1/2 ;

% defining the size of final pdf graphic

% (whitespace and all)

currentpicture := currentpicture xsized 6.6cm ;

3 Celtic patterns

My first Celtic MetaPost pattern was Solomon’s knot:

Intricate knotwork patterns seem to have been popu-
lar all over northern Europe during the Migration era,
but we know them especially well from 9th century
manuscripts like the Book of Kells. These patterns,
as well as spirals and key patterns, are used both for
page borders and to fill whole pages in these opulent
manuscripts.

I started to draw Celtic patterns with MetaPost
partly to create borders and headers for e.g. greeting
cards and partly because I’m intrigued by all types
of braiding structures. I’m slowly trying to create
a set of “building blocks” that can be repeated to
create frames and page backgrounds.

3.1 Celtic keys

I very much wanted to make a Celtic key pattern of
the type that was used for fillings. It is a deceptively
simple design with lines and a few filled shapes (Sloss,
1997, p. 36). First I planned to create the pattern
as I was drawing with a pen, doing whole lines, but
then I finally divided the pattern in five rectangular
blocks — two corners, two sides, one middle — that
can be used to build a larger key pattern picture. As
a result of this approach, some lines consist of several

smaller lines, but that doesn’t show unless you start
editing the graphic. Perhaps not the neatest possible
solution, but it makes “growing” the pattern fairly
straightforward.

% defining line width

numeric w ; w := 3/4 ;

linecap := squared ;

linejoin := mitered ;

% bottom left corner, size 4x4

picture blcorner ;

blcorner := image (

filldraw (0,0) -- (0,4) -- (4,0) -- cycle

withpen pencircle scaled w;

draw (1,1) -- (4,4) withpen pencircle scaled w;

);

% top right corner, size 4x4

picture trcorner ;

trcorner := image (

draw blcorner rotated 180 ; ) ;

% horizontal side element, size 8x4

picture hside ;

hside := image (

filldraw (0,0) -- (1,1) -- (2,0) -- cycle

withpen pencircle scaled w;

draw (0,2) -- (1,3) -- (5,-1)

withpen pencircle scaled w;

draw (2,4) -- (6,0) withpen pencircle scaled w;

draw (5,1) -- (8,4) withpen pencircle scaled w;

filldraw (6,0) -- (8,0) -- (6,2) -- (5,1)

-- cycle withpen pencircle scaled w;

draw (5,3) -- (6,4) withpen pencircle scaled w;

);

% vertical side element, size 4x8

picture vside ;

vside := image (

filldraw (0,0) -- (0,2) -- (1,1) -- cycle

withpen pencircle scaled w;

draw (-1,5) -- (3,1) withpen pencircle scaled w;

draw (2,0) -- (4,2) withpen pencircle scaled w;

draw (3,3) -- (2,4) -- (4,6)

withpen pencircle scaled w;

filldraw (0,6) -- (1,5) -- (2,6) -- (0,8)

-- cycle withpen pencircle scaled w;

draw (1,5) -- (4,8) withpen pencircle scaled w;

) ;

% top left corner, size 4x4

picture tlcorner ;

tlcorner := image (

filldraw (0,0) -- (1,1) -- (0,2) -- cycle

withpen pencircle scaled w;

filldraw (2,4) -- (3,3) -- (4,4) -- cycle

withpen pencircle scaled w;

draw (-1,5) -- (3,1)

withpen pencircle scaled w;

draw (2,0) -- (4,2)

withpen pencircle scaled w;

) ;

% bottom right corner, size 4x4

picture brcorner ;

brcorner := image (

draw tlcorner rotated 180 ;

) ;

% middle element, size 8x8

picture middle ;

middle := image (

draw (0,0) -- (3,3) -- (6,0)

withpen pencircle scaled w;

draw (0,2) -- (2,4) -- (1,5)

withpen pencircle scaled w;

draw (0,6) -- (1,7) -- (7,1) -- (8,2)

withpen pencircle scaled w;

draw (2,8) -- (5,5) -- (8,8)

withpen pencircle scaled w;

draw (7,3) -- (6,4) -- (8,6)

withpen pencircle scaled w;

draw (5,7) -- (6,8)

withpen pencircle scaled w;

draw (2,0) -- (3,1)

withpen pencircle scaled w;

) ;

% draw bottom left corner (starts at origin)

draw blcorner ;

% draw left side (shift by 8)

draw vside shifted (0,4) ;

draw vside shifted (0,12) ;

% draw top left corner (shift by 4+n*8)

draw tlcorner shifted (0,20) ;

% draw bottom row (shift by -4+n*8)

draw hside shifted (4,0) ;

draw hside shifted (12,0) ;

% draw top row (shift by 4+n*8)

draw hside rotated 180 shifted (12,24) ;

draw hside rotated 180 shifted (20,24) ;

% draw right side (shift by 4+n*8)

draw vside rotated 180 shifted (24,12) ;

draw vside rotated 180 shifted (24,20) ;

% draw bottom right corner (shift by 8+n*8)

draw brcorner shifted (24,4);

% draw top right corner

draw trcorner shifted (24,24);

% draw middle

draw middle shifted (4,4) ;

draw middle shifted (12,4) ;

draw middle shifted (4,12) ;

draw middle shifted (12,12) ;

setbounds currentpicture

to boundingbox currentpicture enlarged 1/4 ;

draw boundingbox currentpicture

withpen pencircle scaled 1/4 ;

currentpicture := currentpicture xsized 7.9cm ;

3.2 Josephine knot

Months after creating Solomon’s knot I decided to go
back to the Celtic knotwork patterns, this time with
a design that is known today as a Josephine knot
(taken from Meehan, 2007, p. 199). You may see it
as consisting of two loops, but careful analysis (and
one false start) made me realize that it in fact boils
down to four elements: the end, the middle, a shorter
curve, a longer curve. The middle is the easiest, a
straight line that isn’t rotated, just shifted; the other
three elements sometimes need to be rotated 180
degrees to fit.

Just as with the mosaic floor, I defined each knot
element separately, then shifted the original elements
into correct positions in the knot. After that I created
the rotated elements one at a time and shifted each
into position. When the picture turned into a mess of
white lines because I’d shifted something in a wrong
direction, I found that colouring the newest (top)
element with another colour helped me see what
needed to go where in relation to the elements I’d
placed earlier.

The Celtic scribes used a strict system of grid
points, divisions and sub-divisions to work out their
knots. I stuck to their idea of dividing areas by 4,
16 and 64, but did the final placements by trial and
error rather than by doing the math. Clearly I’ll have
to work out the rules before I can start expanding
the knot e.g. by adding more curves and middles
between the ends, but this simple knot was fairly
easy to work out with estimates partly based on my
earlier experience.

linecap := squared ;

% end piece/loop

picture ending ;

ending := image (

draw (65/64,1/64) -- (1,0) -- (0,0)

-- (0,3) -- (1,3) -- (2,2) ; );

% the longer curve piece

picture lcurve ;

lcurve := image (

draw (1/64,2+1/64) -- (0,2) -- (0,1)

-- (1,0) -- (2,0) -- (2+1/64,1/64) ; );

% the shorter curve piece

picture scurve ;

scurve := image (

draw (0,1) -- (1,0) -- (2,0)

-- (2+1/64,1/64) ; );

% the straight line in the middle

picture dline ;

dline := image (

draw (1/2,1/2) -- (3/2,3/2) ; ) ;

% only this pen width works with these settings!

drawoptions (withpen pencircle

scaled 5/8 withcolor white ) ;

draw ending shifted (1/2,1/2) ;

draw lcurve shifted (3/2,1/2) ;

draw scurve shifted (7/2,1/2) ;

draw dline shifted (2,1);

draw ending rotated 180 shifted (15/2,7/2) ;

draw scurve rotated 180 shifted (9/2,7/2) ;

draw dline shifted (4,1) ;

draw lcurve rotated 180 shifted (13/2,7/2);

setbounds currentpicture

to boundingbox currentpicture enlarged 1/8 ;

addbackground withcolor black ;

currentpicture := currentpicture xsized 6cm ;

4 References

I’m indebted to Hans Hagen, both for the patience
with my very basic questions and for the invaluable
snippets of code that have pushed me into new paths
of thinking.

4.2 Online resources

Greek keys:
http://gwydir.demon.co.uk/jo/greekkey/

On Romano-British mosaics:
http://www.asprom.org/resources/factsheets/

ASPROM_factsheet_Romano-British_mosaics.

pdf

The mosaic floors at Hospitalia at Hadrian’s Villa:
http://commons.wikimedia.org/wiki/File:

Pavement_Hospitalia_Villa_Hadriana_n2.jpg

http://commons.wikimedia.org/wiki/Villa_

Adriana#Hospitalia_and_Imperial_Triclinium

On the Book of Kells:
http://en.wikipedia.org/wiki/Book_of_Kells

More on my Celtic knots:
http://www.lucet.fi/craftex/celtic-metapost

� Mari Voipio
mari dot voipio (at) lucet dot fi

http://www.lucet.fi

Recreating historical patterns with MetaPost


